# Copyright (c) 2009-2011 XORP, Inc and Others
#
# This program is free software; you can redistribute it and/or modify
# it under the terms of the GNU General Public License, Version 2, June
# 1991 as published by the Free Software Foundation. Redistribution
# and/or modification of this program under the terms of any other
# version of the GNU General Public License is not permitted.
#
# This program is distributed in the hope that it will be useful, but
# WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. For more details,
# see the GNU General Public License, Version 2, a copy of which can be
# found in the XORP LICENSE.gpl file.
#
# XORP Inc, 2953 Bunker Hill Lane, Suite 204, Santa Clara, CA 95054, USA;
# http://xorp.net

# $XORP$

import os
Import('env')

env = env.Clone()

env.AppendUnique(CPPPATH = [
    '#',
    '$BUILDDIR',
     ])

env.AppendUnique(LIBPATH = [
    '$BUILDDIR/libxorp',
    '$BUILDDIR/libcomm',
    '$BUILDDIR/libxipc',
    '$BUILDDIR/libproto',
    '$BUILDDIR/xrl/interfaces',
    '$BUILDDIR/xrl/targets',
    '.',
    ])

env.AppendUnique(LIBS = [
    'xif_rip',
    'xif_ripng',
    'xorp_ipc',
    'xorp_core',
    'xorp_proto',
    'xorp_comm',
    ])

if (env.has_key('mingw') and env['mingw']):
    env.Append(LIBS = [
        'iphlpapi',
#        'mprapi',
        'regex',
        'winmm',
        ])

    env.Append(LIBS = ['xorp_core', 'crypto', 'ws2_32'])


env.Replace(RPATH = [
    env.Literal(env['xorp_tool_rpath'])
])

shpeersrcs = [
	'show_peer_stats.cc',
	'common.cc'
    ]

shstatssrcs = [
	'show_stats.cc',
	'common.cc'
    ]

shpeer = env.Program(target = 'rip_show_peer_stats', source = shpeersrcs)
shstats = env.Program(target = 'rip_show_stats', source = shstatssrcs)
if env['enable_builddirrun']:
    for obj in shpeer:
        env.AddPostAction(shpeer,
            env.Copy(obj.abspath,
                        os.path.join(env['xorp_alias_tooldir'], str(obj))))
    for obj in shstats:
        env.AddPostAction(shstats,
            env.Copy(obj.abspath,
                        os.path.join(env['xorp_alias_tooldir'], str(obj))))
env.Alias('install', env.InstallProgram(env['xorp_tooldir'], shpeer))
env.Alias('install', env.InstallProgram(env['xorp_tooldir'], shstats))

Default(shpeer, shstats)
